

Lisp in Summer Projects Submission

Submission Date	2013-10-20 07:43:10
Full Name	Dogadailo Dmytro Victorovich
Country	Ukraine
Project Name	MetaJS (knowledge-oriented programming language)
Type of software	command-line/terminal app
General category	lisp compiler/interpreter
LISP dialect	other
GitHub URL	https://github.com/dogada/metajs/tree/LISP
Did you start this project?	Yes, all the code is written by me
Project Description	I want to upload a free-form 3-4 page PDF composition.
Upload 3-4 page detailed PDF	metajs_lisp.pdf
Build Instructions	<p>MetaJS requires Node.Js for development environment. For installation instructions please look at https://github.com/dogada/metajs/tree/LISP#how-to-install-and-try-metajs</p> <p>Once you have MetaJS installed, please run `make` in the root project folder. MetaJS will rebuild itself and run test suite.</p>
Test Instructions	<p>To run unit tests please run:</p> <pre>\$ make test</pre> <p>or</p> <pre>\$ metajs -x test/index.mjs</pre> <p>To check unit tests for errors please use:</p> <pre>\$ metajs --lint test/index.mjs</pre>
Execution Instructions	<p>To start a REPL:</p> <pre>\$ metajs</pre> <p>To execute a MetaJS file:</p>

```
$ metajs -x test/index.mjs
```

To compile a file to stdout:

```
$ metajs test/logos.mjs
```

Describe any bugs or caveats

MetaJS is in alpha stage but can recompile itself already. For typical use cases it should work without errors. Compiler not always clearly report found errors unfortunately.

Screen shots

□ [metajs_logos.png](#)

```
dogada@i3:MJS$ metajs
metajs 0.1.1 TEPL (Translate Eval Print Loop)
metajs> (defn show-timeline (username limit:10) #"Top $limit entries in $username's timeline.")
metajs> (show-timeline "John")
'Top 10 entries in John's timeline.'
metajs> (show-timeline "John" 5)
'Top 5 entries in John's timeline.'
metajs> (show-timeline)
<unknown>:1:1 Error(2): username is required for show-timeline
metajs> (let username "John" (show-timeline))
'Top 10 entries in John's timeline.'
metajs> (let username "John" limit 3.14 (show-timeline))
'Top 10 entries in John's timeline.'
metajs> (def actors ["Neo" "Trinity" "Morpheus"])
metajs> (def [neo trinity morpheus] actors)
metajs> [neo trinity morpheus]
[ 'Neo', 'Trinity', 'Morpheus' ]
metajs> (defmacro when (x & code) `(if -x (do -@code)))
metajs> (let earth-moves true (when earth-moves "And yet it moves!"))
'And yet it moves!'
metajs> (process .hrtime @1)
172959867
metajs> (+ 2 2)
4
metajs> (eval (compile (read "(+ 2 2)")))
4
metajs> Bye!
```

[metajs_repl.png](#)

Official

I have read rules and have abided by them.

I am 18 years of age or older.

I am not living in Brazil, Quebec, Saudi Arabia, Cuba, Iran, Myanmar (Burma), North Korea, Sudan, or Syria.

MetaJS

Knowledge-oriented programming language

Purpose

MetaJS is knowledge-oriented programming language. MetaJS compiler can generate source code, intentionally missed by programmer. This approach leads to loosely coupled and highly dynamic programs. MetaJS is a dialect of Lisp and provides many of the typical Lisp features including powerful macro system. It compiles to Javascript without runtime dependencies.

Function

Typical usage scenario: signature of a library function is changed but all arguments for new version of the function can be found in the call context. MetaJS can adapt all calls of such function to the new signature automatically. A programmer starts to use less nouns (local variables and function arguments) and more verbs (function calls). MetaJS compiler finds what does the programmer mean from semantic model of the program and surrounding context.

Motivation

Today compilers analyze only the grammar of the language and ignore the semantics of each specific program. As a result, we get the source code with high level of redundancy. With the increasing size of the program this naturally leads to difficulties in its support. Major changes in the application source code become virtually impossible without causing of new bugs and regressions.

Audience

The complexity of software and user's expectations increase much faster than programmer's productivity. MetaJS attempts to close this gap and offers to small teams and individual programmers a tool that allows to do more for less time. Semantic-aware compiler orthogonally combines with traditional Lisp meta-programming and provides new experience for all programmers who value own time.

Methodology

In addition to traditional macros that transform program at read- and compile-time MetaJS uses semantic transformations that runs after read- and compile-macros. In MetaJS functions can have required and optional parameters. When MetaJS compiler finds a function call without a required parameter it attempts to “resolve” it to make the valid function call.

The compiler tries to find all symbols and compound forms that can be used instead of missed function argument. The compiler replaces missed function argument only when it finds exactly one valid substitution. When valid substitution is not found or found more than one, compiler reports an error that programmer need to resolve manually. The same approach is used in the version control systems. Most of time merges of different code branches are done automatically, but if there are a merge conflict – programmer need to resolve it manually or use different merge algorithm.

MetaJS uses two classes of semantic transformation: symbolic and entitative. During symbolic transformations compiler uses symbols available in the lexical scope of the function call instead of missed argument of the function. To qualify a symbol for the substitution of the missed function argument MetaJS tries to match name and meta information of the missed argument and each symbol available in the lexical scope of incomplete function call. Name of the argument can be matched with meta of context symbol and vice versa forming four possible types of combinations.

Let's see an example of symbolic transformation when name of missed function argument is matched with meta type of context symbol. On the left is MetaJS code and on the right is JavaScript code generated from MetaJS code. Result of semantic transformation is highlighted with **red** color. You can check this and next examples right inside your browser on the <http://metajs.coect.net/>.

<pre>(defn get-user-timeline (username) #"\$username timeline")</pre>	<pre>var getUserTimeline = (function(username) { return (" + username + " timeline"); });</pre>
<pre>(let* (u:username "dogada") (get-user-timeline))</pre>	<pre>(function(u) { return getUserTimeline(u); })("dogada");</pre>

Required argument **username** of the function **get-user-timeline** is missed in the function call. MetaJS uses local variable **u**, because it's declared with meta type **username**. The code **u:username** defines symbol **u** and declares that it is related to something called **username**. Meta types are not need to be defined at all. Compiler uses meta types and entities to build semantic model of the program.

Entities are defined with (**entity**) macro and used solely by compiler to understand relations between symbols. Entities are not types in statically-typed languages sense and have no corresponding JavaScript code. Entities are main building block of semantic model of the program.

Let's see an example of an entitative transformation. Such transformations always use an explicitly defined entity to find a valid relation between symbols available in the lexical scope of function call and missed function argument. Entities may define simple relation like property relation, for example (**has username**). Custom relation able to emit any form instead of missed argument, for example (**rel username `(. ~sym 'session ~rel)**), where **sym** is actual symbol matched to the entity and **rel** is target itself.

<pre>(entity request "Entity of demo web request." (has username))</pre>	
<pre>(defn get-user-timeline (user-id:username) #"\$user-id timeline")</pre>	<pre>var getUserTimeline = (function(userId) { return (" + userId + " timeline"); });</pre>
<pre>(let* (req:request {username: "dogada"}) (get-user-timeline))</pre>	<pre>(function(req) { return getUserTimeline(req.username); })({username: "dogada"});</pre>

Required argument **user-id** of the function **get-user-timeline** is missed in the function call. MetaJS uses property **username** of local variable **req**, because it's declared with meta type **request** and entity **request** declares relation between **request** and **username** entities and **user-id** argument declared with meta type **username**.

You can find examples of all 8 types of semantic transformations used by MetaJS in the http://metajs.coect.net/pdf/metajs_semantic_code_transformations.pdf.

Conclusion

MetaJS is implemented in MetaJS, so even if it's still in alpha stage it's quite feature complete already. Famous Lisp macro system allows to add more features without need to change language itself. Anyone can try it without leaving a browser on <http://metajs.coect.net/> or install as NodeJs module.

It's easy to predict that in large programs symbols and entities with same name may have different meaning in different source files even inside single project. When you import external libraries, you may expect even more name conflicts. MetaJS plans to solve this problem with namespaces and using of fully qualified names of entities inside compiler. A programmer can import and alias entities from other modules exactly like symbols are imported between namespaces.

Next obvious area of improvement is meta types inheritance and various checks of code validity based on semantic model of program.

The final goal is to make a compiler that can learn programmer's style at least particularly. A programmer concentrates own efforts on domain models design and architecture of the program. A programmer tells to MetaJS compiler what to do and teach it to understand program's entities and relations between symbols. Computer in real-time mode analyzes semantic model of the program, compares it with the code that programmer types and provides immediate feedback rather than shows blinking cursor.

Screenshots, REPL and logos:

```
dogada@i3:MJS$ metajs
metajs 0.1.1 TEPL (Translate Eval Print Loop)
metajs> (defn show-timeline (username limit:10) #"Top $limit entries in $username's timeline.")
metajs> (show-timeline "John")
'Top 10 entries in John\'s timeline.'
metajs> (show-timeline "John" 5)
'Top 5 entries in John\'s timeline.'
metajs> (show-timeline)
<unknown>:1:1 Error(2): username is required for show-timeline
metajs> (let username "John" (show-timeline))
'Top 10 entries in John\'s timeline.'
metajs> (let username "John" limit 3.14 (show-timeline))
'Top 10 entries in John\'s timeline.'
metajs> (def actors ["Neo" "Trinity" "Morpheus"])
metajs> (def [neo trinity morpheus] actors)
metajs> [neo trinity morpheus]
['Neo', 'Trinity', 'Morpheus']
metajs> (defmacro when (x & code) `(if ~x (do ~@code)))
metajs> (let earth-moves true (when earth-moves "And yet it moves!"))
'And yet it moves!'
metajs> (process .hrttime @1)
172959867
metajs> (+ 2 2)
4
metajs> (eval (compile (read "(+ 2 2)")))
4
metajs> Bye!
```

```
(defn check-call (name args)
  (def name* (token-value* name)
    expected (find-fn-signature name*))
  (check-call-params name args expected))

(defn entity-resolve (arg)
  "Use entities associated with arg to find substitution for missed argument."
  (def entities (find-entities arg.token)
    bound (filter (map entities (f (e) {entity: e symbols: (find-symbols [e.name])}))
      (f (es) (not-empty? es.symbols))))
  (merge (map bound (f (es)
    (map es.symbols #(es.entity.code (. % 'name)
      ['quote es.entity.token]))))))

(defn symbol-resolve (arg)
  "Find in the current lexical scope symbols with arg's name or meta type."
  (map (find-symbols (get-token-entities arg.token)) #(get % 'name)))

(defn report-resolve-error (arg fn-token forms)
  "Show all possible candidates for resolving the missed argument of a function."
  (def name arg.name
    fn-name (token-value* fn-token)
    resolvers ((map forms compile-one) .join ", "))
  (dump-scope-logos)
  (lint-many-candidates #"Too many candidates for $name in $fn-name: $resolvers." fn-token)
  null)

(defn resolve-arg (arg fn-token)
  "Replace missed argument of a function with a form or report an error."
  (def forms (concat (symbol-resolve) (entity-resolve))
    name arg.name)
  (if (contains-one? forms) (first forms)
    (contains-many? forms) (report-resolve-error)))

(defmacro entity (name & rels)
  "Define entity in the current lexical scope."
  (def doc "")
  (when (quoted? (first rels))
    (set [doc rels] (frest rels)))
  ((get-scope) .set-entity name rels doc)
  undefined)
```